

DS MP2I : Arbres

Solution.

□

Toujours indiquer le type et l'objectif des fonctions auxiliaires. Tout code long doit être expliqué.

1 Tableau associatif

Le haché *DJB2* (*Dan Bernstein*) est un algorithme simple de hachage de chaîne de caractères datant de 1991. Il associe un entier positif à toute chaîne de caractères.

On donne, pour l'implémenter, le code suivant :

```
1 | let hash_djb2 (s : string) : int =
2 |   let h = ref 5381 in
3 |   String.iter (fun c ->
4 |     h := (!h lsl 5) + !h + Char.code c
5 |   )
6 |   s;
7 |   !h land max_int;;
```

Dans ce code, `Char.code c` désigne le code ASCII du caractère `c` (ainsi, `Char.code 'a'` vaut 97).

```
1 | # hash_djb2 "toto";;
2 | - : int = 6385734411
```

Question 1.

L'appel `hash_djb2 s` effectue AVANT la dernière ligne un calcul $P(x)$ où P est un certain polynôme dont le degré dépend de n (la longueur de s) et x une valeur fixée. La première partie du code est donc une implémentation de l'algorithme de Horner.

Identifier le polynôme P et la valeur de x .

Solution.

$$P(x) = 5381x^{|s|} + \sum_{i=0}^{|s|-1} \text{code}(s_i)x^{|s|-1-i}$$

Ici $x = 33 = 2^5 + 1$. car

```
1 | !h lsl 5 + !h = !h * (1 lsl 5 + 1)
```

□

Question 2.

Expliquer la dernière ligne : on peut la comprendre comme l'ajout d'une certaine valeur à `!h` si celui-ci est négatif et d'une autre valeur si `!h` est positif. Mais lesquelles ?

Solution. On obtient un nombre positif : on ne garde que les bits de `!h` après le bit de poids fort : le bit de signe de `max_int` est 0 et tous les autres valent 1.

Comme les entiers OCaml sont codés sur 63 bits (le 64ème étant réservé pour stocker des métas-informations par OCaml), l'opération finale revient à ajouter 2^{62} à `!h` lorsqu'il est négatif. Et à ne rien faire (ou à ajouter 0) sinon.

□

On donne les types suivants :

```

1 | type key = string
2 |   (*Un bucket est la collection d'éléments associés à un même indice.*)
3 |   type 'v bucket = (key * 'v) list
4 |   type 'v sht = {
5 |     mutable buckets : 'v bucket array;
6 |     hash : key -> int;
7 |     mutable size : int; (* nombre de seaux *)
8 |   }
9 |   (* Seuil de charge à partir duquel on redimensionne *)
10 | let max_load_factor = 2.0 /. 3.0

```

Question 3.

Écrire la fonction

`create (hash : 'a -> int) (initial_capacity : int) : 'a sht`. Elle prend en paramètre une fonction `hash` de hachage et une indication de la capacité du tableau de stockage. Elle renvoie un dictionnaire vide. Si le second paramètre est inférieur à 1, on n'en tient pas compte et le tableau de stockage prend une taille de 1.

Solution. Code

```

1 | let create hash initial_capacity =
2 |   let cap = max 1 initial_capacity in
3 |   { buckets = Array.make cap []; size = 0; hash }

```

□

Lorsque la table devient trop pleine, on double la capacité, puis on réinsère chaque élément dans la nouvelle table. La constante `max_load_factor` est une variable globale telle que le redimensionnement est déclenché si le ratio nombre d'associations / capacité est supérieur à cette quantité. On prend $\frac{2}{3}$ comme en Python¹ pour cette valeur.

Question 4.

Écrire la fonction `resize (tbl : 'a sht) : unit` qui double la taille du tableau de stockage et réinsère les associations dans l'ordre où elles étaient auparavant.

Solution. Code

```

1 | (* Redimensionnement : double la capacité et réinsère tous les
2 |   éléments, en préservant l'ordre de chaque bucket *)
3 | let resize tbl =
4 |   let old = tbl.buckets in
5 |   let new_cap = max 1 (Array.length old * 2) in
6 |   let new_buckets = Array.make new_cap [] in
7 |   Array.iter (fun bucket ->
8 |     (* On parcourt chaque bucket à l'envers pour conserver l'ordre
9 |       original lors des réinsertions *)
10 |     List.iter (fun (k, v) ->
11 |       let idx = (tbl.hash k) mod new_cap in
12 |       new_buckets.(idx) <- (k, v) :: new_buckets.(idx)
13 |     ) (List.rev bucket)
14 |   ) old;
15 |   tbl.buckets <- new_buckets

```

□

1. En OCaml c'est plutôt $\frac{1}{2}$.

Question 5.

Écrire la fonction `add (tbl : 'a sht) (k : key) (v : 'a) : unit` qui ajoute une association au dictionnaire. Éventuellement, si le ratio taille/capacité est trop élevé, on effectue d'abord un redimensionnement. La fonction `add` se contente d'ajouter une association `(k,v)`. Il est possible qu'une autre association de même clé se trouve plus loin dans le seau.

Solution. Code

```
1 | (* Ajout d'une paire clé/valeur, avec redimensionnement si
2 |   nécessaire *)
3 | let add tbl key value =
4 |   let cap = Array.length tbl.buckets in
5 |   if float tbl.size >= max_load_factor *. float cap then resize tbl;
6 |   let idx = (tbl.hash key) mod (Array.length tbl.buckets) in
7 |   tbl.buckets.(idx) <- (key, value) :: tbl.buckets.(idx);
8 |   tbl.size <- tbl.size + 1
```

□

Question 6.

Écrire la fonction `find (tbl : 'a sht) (k:key) : 'a` qui renvoie la valeur associée à la clé passée en paramètre et soulève `Not_found` sinon. Si plusieurs associations de même clé se trouvent dans le seau, alors `find` renvoie la valeur de la première rencontrée (donc la plus récemment ajoutée) sans se préoccuper des autres (celles qui sont plus loin dans le seau).

Solution. Code

```
1 | (* Recherche d'une valeur par clé, lève Not_found si absente *)
2 | let find tbl key =
3 |   let idx = (tbl.hash key) mod (Array.length tbl.buckets) in
4 |   List.assoc key tbl.buckets.(idx)
```

□

2 Peignes

On définit le type arbre de la manière suivante :

```
1 | type arbre =
2 |   | Feuille of int
3 |   | Noeud of arbre * arbre ;;
```

On dit qu'un arbre est un *peigne* si tous les nœuds internes (à l'exception éventuelle de la racine) ont au moins une feuille pour fils. On dit qu'un peigne est un peigne *strict* si sa racine a au moins une feuille pour fils, ou s'il est réduit à une feuille. On dit qu'un peigne est *rangé* si le fils droit d'un nœud interne est toujours une feuille. Un arbre réduit à une feuille est considéré comme un peigne rangé.

Question 7.

Représentation.

1. Dessiner un peigne rangé à 5 feuilles;
2. puis donner son code OCaml.

Solution. On a





FIGURE 1 – Un peigne à cinq feuilles



Et voici son codage en OCAML :

```
1 || let r = Noeud(Noeud(Noeud(Noeud(Feuille 5,Feuille 4),
2 ||     Feuille 3),Feuille 2),Feuille 1));
```

□

Question 8.

Quelle est la hauteur d'un peigne rangé à n feuilles ? On justifiera la réponse.

Solution.

Remarque. Dans vos rédactions, indiquez bien à quel endroit vous appliquez l'hypothèse d'induction ou de récurrence.

Par Induction Pour un arbre A on note $h(A)$ sa hauteur et $f(A)$ son nombre de feuilles.

Cas de base pour l'arbre feuille : OK.

Soit A un peigne rangé dont le fils gauche G vérifie l'hypothèse d'induction.

Alors $h(G) = f(G) - 1$ (par HI). Or $f(A) = f(G) + 1$ et

$$h(A) = \max(h(G), 1) + 1 = h(G) + 1 = f(G) - 1 + 1 = f(G) = f(A) - 1$$

□

Question 9.

Ecrire une fonction `est_range : arbre → bool` qui renvoie `true` si l'arbre donné en argument est un peigne rangé.

Solution. Code

```
1 || let rec est_range a =
2 ||     match a with
3 ||     | Feuille _ -> true
4 ||     | Noeud(g, Feuille _) -> est_range g
5 ||     | _ -> false;;
```

□

Question 10.

Caractérisation des peignes.

1. Écrire une fonction `est_peigne_strict : arbre → bool` qui renvoie `true` si l'arbre donné en argument est un peigne strict.
2. En déduire une fonction `est_peigne : arbre → bool` qui renvoie `true` si l'arbre donné en argument est un peigne.

Solution. Voici

```

1 | let rec est_peigne_strict a =
2 |   match a with
3 |   | Feuille _ -> true
4 |   | Noeud(g, Feuille _) -> est_peigne_strict g
5 |   | Noeud(Feuille _, d) -> est_peigne_strict d
6 |   | _ -> false;;
7 |
8 | let est_peigne a =
9 |   match a with
10 |  | Feuille _ -> true
11 |  | Noeud(g, d) -> est_peigne_strict g && est_peigne_strict d;;

```

□




FIGURE 2 – Une rotation

On souhaite maintenant ranger un peigne donné. Supposons que le fils droit D de sa racine ne soit pas une feuille. Notons A_1 le sous-arbre gauche de la racine, f l'une des feuilles du noeud D et A_2 l'autre sous-arbre du noeud D . On va utiliser l'opération de *rotation* qui construit un nouveau peigne où


- le fils droit de la racine est le sous-arbre A_2 ;
- le fils gauche de la racine est un noeud de sous-arbre gauche A_1 et de sous-arbre droit la feuille f .

Question 11.


Application de rotations.

1. Dessiner le résultat d'une rotation sur l'arbre de la figure 1
2. Appliquer encore une rotation.

Solution. Après une rotation :



Après deux rotations :



□

Question 12.

Ecrire une fonction `rotation : arbre → arbre` qui effectue une rotation à la racine de l'arbre. La fonction renverra l'arbre initial si une rotation n'est pas possible.

Solution. Pas de difficulté, mais il faut faire un filtrage un peu détaillé selon que le fils droit de la racine à une feuille à droite ou à gauche :

```

1 | let rotation a =
2 |   match a with
3 |   | Noeud (a1, Noeud (a2, Feuille x)) ->
4 |     Noeud(Noeud (a1, Feuille x),a2)
5 |   | Noeud (a1, Noeud (Feuille x, a2)) ->
6 |     Noeud(Noeud (a1, Feuille x),a2)
7 |   | _ -> a ;;

```

□

Question 13.

Ecrire une fonction `rangement : arbre → arbre` qui range un peigne donné en argument, c'est à dire qu'il renvoie un peigne rangé ayant les mêmes feuilles que celui donné en argument. La fonction renverra l'arbre initial si celui-ci n'est pas un peigne.

Solution. Code

```

1 | let rangement a =
2 |   (* pour prévenir les boucles infinies *)
3 |   if not (est_peigne a) then a
4 |   else
5 |     let rec _ranger a = match a with
6 |       | Feuille _ -> a
7 |       | Noeud (g, (Feuille _ as d)) -> Noeud (_ranger g, d)
8 |       | _ -> _ranger (rotation a)
9 |     in
10 |    _ranger a;;

```

□

3 Centre et diamètres

On donne les types

```

1 | (* Définition de l'arbre binaire *)
2 | type 'a tree =
3 |   | Empty
4 |   | Node of 'a * 'a tree * 'a tree
5 |
6 | (* Pour représenter le centre de l'arbre *)
7 | type 'a center =
8 |   | Single of 'a (* centre unique *)
9 |   | Double of 'a * 'a (* deux centres *)

```

Dans cet exercice, on fixe la *hauteur* du nœud vide à 0. La *hauteur* d'un nœud est la longueur (exprimée en nombre de nœuds) du chemin le plus long entre ce nœud et les feuilles qui en descendent.

Pour définir un *chemin*, on considère les arbres comme des graphes non orientés : un chemin se parcourt dans les deux sens (il n'y a plus de père ni de fils seulement des *voisins*). Dans ce contexte, un *diamètre* d'un arbre est un plus long chemin entre deux nœuds de l'arbre.

Question 14.

Établir qu'un arbre peut avoir plusieurs diamètres. C'est à dire qu'il existe deux chemins n'empruntant pas les mêmes sommets et dont la longueur est maximale.

Solution. Dans un arbre parfait de hauteur ≥ 2 , les chemins des deux feuilles les plus à gauche vers les deux feuilles les plus à droite sont des diamètres. Il y a donc au moins 4 diamètres □

Question 15.

Établir qu'il existe un arbre dont aucun diamètre ne passe par la racine.

Solution. Ajoutons une racine qui aurait pour fils gauche l'arbre de la figure 3 et pas de fils droit. □

Alors aucun diamètre ne passe par la racine.

Un *centre* d'un arbre est un nœud tel que la distance maximale (en nombre de nœuds) entre ce nœud et tout autre nœud de l'arbre est minimal (cf figure 3).




FIGURE 3 – Le diamètre est 6, le centre est le nœud d'étiquette 3

Listing 1 – Implémentation de l'arbre de la figure 3

```

1 let example_tree =
2   Node (1,
3     Node (2,
4       Node (4, Empty, Empty),
5       Node (5, Empty, Empty)
6     ),
7     Node (3,
8       Empty,
9       Node (6, Node (7, Empty, Node (8, Empty, Empty)), Empty)
10    )
11  );;
```

Une propriété bien connue est que tout arbre possède soit un centre unique (si la longueur d'un diamètre est impaire), soit exactement deux centres adjacents (si la longueur d'un diamètre est paire). Selon la parité de la longueur, le centre est le point au milieu d'un diamètre ou les deux points au milieu.

Remarque. Cette propriété se prouve en considérant qu'un arbre est un graphe non orienté connexe et acyclique et en considérant la suite des graphes obtenus en partant de l'arbre initial et en supprimant toutes les feuilles du graphe précédent.

Une approche classique pour trouver le centre consiste à :

- Calculer le diamètre de l'arbre, c'est-à-dire le chemin le plus long (en nombre de nœuds) reliant deux nœuds quelconques.
- Le ou les centres se trouvent alors au milieu de ce chemin. Plus précisément, si le chemin comporte un nombre impair de nœuds, le centre est unique (le nœud médian) ; s'il comporte un nombre pair de nœuds, il y a deux centres (les deux nœuds centraux).

Pour le calcul du diamètre, on utilise une approche par programmation dynamique qui, pour chaque nœud, calcule :

- La longueur (en nombre de nœuds) de la plus longue branche descendant du nœud (la hauteur) et le chemin correspondant.
- Un diamètre et sa longueur, ce diamètre pouvant être soit dans le sous-arbre gauche, soit dans le sous-arbre droit, soit passer par le nœud courant (en combinant la plus grande branche gauche et la plus grande branche droite).

Question 16.

Écrire une fonction `diameter (t: 'a tree) : int * 'a list * int * 'a list` qui prend en paramètre un arbre et renvoie un quadruplet (`height`, `height_path`, `diam`, `diam_path`) où :

- `height` désigne la hauteur (nombre de nœuds) du chemin descendant le plus long partant de `t` ;
- `height_path` est le chemin correspondant (du nœud courant jusqu'à la feuille la plus éloignée) donné sous forme de liste d'étiquettes ;
- `diam` désigne le diamètre (en nombre de nœuds) de l'arbre `t` ;
- `diam_path` est le chemin correspondant au diamètre donné sous forme de liste d'étiquettes.

```
1 | # diameter example_tree;;
2 | - : int * int list * int * int list =
3 | (5, [1; 3; 6; 7; 8], 7, [4; 2; 1; 3; 6; 7; 8])
```

On privilégiera une approche EN UN SEUL PARCOURS.

Solution. Code

```
1 | let rec diameter t =
2 |   match t with
3 |   | Empty -> (0, [], 0, [])
4 |   | Node(v, left, right) ->
5 |     let (hl, hpath_left, diam_left, diam_path_left) = diameter left in
6 |     let (hr, hpath_right, diam_right, diam_path_right) = diameter right in
7 |     (* Calcul de la hauteur et du chemin descendant
8 |        le plus long depuis le noeud courant *)
9 |     let (h, hpath) =
10 |       if hl >= hr then (hl + 1, v :: hpath_left)
11 |       else (hr + 1, v :: hpath_right)
12 |     in
13 |     (* Candidat pour un diamètre passant par le noeud courant *)
14 |     let candidate_diam = hl + hr + 1 in
15 |     let candidate_path =
16 |       let left_path = if left = Empty then [] else List.rev hpath_left in
17 |       let right_path = if right = Empty then [] else hpath_right in
18 |       left_path @ [v] @ right_path
19 |     in
20 |     (* Choix du meilleur diamètre *)
21 |     let (best_diam, best_path) =
22 |       if candidate_diam >= diam_left && candidate_diam >= diam_right then
23 |         (candidate_diam, candidate_path)
24 |       else if diam_left >= diam_right then
25 |         (diam_left, diam_path_left)
26 |       else
27 |         (diam_right, diam_path_right)
28 |     in
29 |     (h, hpath, best_diam, best_path);;
```

□

Question 17.

Écrire une fonction `center (t: 'a tree) : int list` qui, étant donné un arbre binaire, retourne l'étiquette (ou le couple d'étiquettes) du centre (ou des deux centres) sous la forme d'une valeur de type `centre`.

```
1 | # center example_tree;;
2 | - : int center = Single 3
```

Solution. Code

```
1 |
2 |
3 | let rec tronque k l =(*enlève les k premiers éléments*)
4 |   match l, k with
5 |   | l, 0 -> l
6 |   | [], _ -> []
7 |   | _ :: q, k when k > 0 -> tronque (k - 1) q
8 |   | _ -> invalid_arg "tronque"
9 | ;;
10 |
11 |
12 | (*V1*)
13 | let center t =
14 |   let (_, _, n, diam_path) = diameter t in
15 |   if n = 0 then failwith "Arbre vide"
16 |   else
17 |     match n mod 2, tronque ((n - 1) / 2) diam_path with
18 |     | 1, x :: _ -> Single x
19 |     | 0, c1 :: c2 :: _ -> Double (c1, c2)
20 |     | _ -> failwith "center"
21 | ;;
22 | (*
23 | paramètre de tronque :
24 | si n impair, (n - 1) / 2 = n / 2
25 | si n pair, (n - 1) / 2 = (n / 2) - 1
26 | *)
27 |
28 |
29 | (*V2 en parcourant le diamètre et son inverse simultanément*)
30 | let center t =
31 |   let (_, _, _, diam_path) = diameter t in
32 |   let rev_diam_path = List.rev diam_path in
33 |
34 |   let rec aux l1 l2 =
35 |     match l1, l2 with
36 |     | [], [] -> failwith "Arbre vide"
37 |     | [x], [x'] when x = x' ->
38 |       Single x
39 |     | x :: y :: q, x' :: y' :: q' ->
40 |       if x = x' then
41 |         Single x
42 |       else if x = y' then
43 |         Double (x, y)
44 |       else
45 |         aux (y :: q) (y' :: q')
46 |     | _ ->
47 |       failwith "diamètre mal formé"
48 |   in
49 |   aux diam_path rev_diam_path
50 | ;;
```

□